﻿package  {
	
	import flash.display.*;
	import flash.events.*;
	import flash.media.Sound;
	import flash.net.URLRequest;
	import flash.display.Loader;
	import flash.media.SoundChannel;
	
	public class Jokua extends MovieClip {
		
		//var bakarrik jartzen dugu, honela atributuak heredatu ahal izateko.
		
		//joku guztiek dituzten irudiak
		var atzera:Botoia; //menura itzultzeko botoia
		var fondoa:Irudia; 
		var galderaIkurra:Botoia; //azalpena ikusteko botoia
		var azalpena:Irudia;
		var zuzenak:int=0;
		
		//soinuak
		var ondo:Sound;
		var txarto:Sound;
		var channel:SoundChannel;
		
		//atributuak
		var haur:Boolean; //haur jolasetan edo lehen jolasetan gauden jakiteko.
		var izena:String; //jokuaren izena
		
		
		public function Jokua(){
			
			fondoaGehitu();
			botoiakHasieratu();
			soinuakHasieratu();

		}
		
		public function fondoaGehitu(){
			fondoa = new Irudia("irudiak/fondoak/liburua.png");
			addChildAt(fondoa,0);
		}
		
		public function irabaziMezuaErakutsi(){
			
			var mezua:Mezua;
			mezua = new Mezua();	
			mezua.x=0;
			mezua.y=0;
			addChild(mezua);//Ikusi ahal izateko beharrezkoa
			mezua.addEventListener(MouseEvent.CLICK,menuraBueltatu);
			
		}
		
		public function emaitzaErakutsi(j:int, z:int){
			
			var e:Emaitza= new Emaitza(j,z);
			elementuakEzabatu(this,5);
			e.x=100;
			e.y=100;
			addChild(e);
			e.addEventListener(MouseEvent.CLICK,menuraBueltatu);
			
		}
		
		private function botoiakHasieratu(){
			atzera = new Botoia("irudiak/botoiak/atzera.png");
			atzera.setArgiKolorea("urdina");
			atzera.x=900;
			atzera.y=590;
			addChildAt(atzera,1);
			atzera.addEventListener(MouseEvent.CLICK,menuraBueltatu);
			
			galderaIkurra = new Botoia("irudiak/botoiak/azalpena.png");
			galderaIkurra.setArgiKolorea("horia"),
			galderaIkurra.x=900;
			galderaIkurra.y=40;
			addChildAt(galderaIkurra,2);
			galderaIkurra.addEventListener(MouseEvent.CLICK,azalpenaErakutsi);
		}
		
		private function soinuakHasieratu(){
			ondo = new Sound(new URLRequest("Soinuak/ondo.mp3"));
			txarto = new Sound(new URLRequest("Soinuak/txarto.mp3"));
		}
		
		private function menuraBueltatu(eventObject:MouseEvent){
			
			x=0;
			y=0;
			
			elementuakEzabatu(this,0);
			if (haur){
				var menua1:HaurMenua;
				menua1 = new HaurMenua;
				addChild(menua1);	
			} else {
				var menua2:LehenMenua;
				menua2 = new LehenMenua;	
				addChild(menua2);	
			}
			
			
		}
		
		public function elementuakEzabatu(clip, i:int){
			
			var l:int = clip.numChildren-1;
			for(var j = l; j >= i;j--){
				clip.removeChildAt(j);
			}
		}
		
		
		private function azalpenaErakutsi(eventObject:MouseEvent){
			azalpena = new Irudia("irudiak/azalpenak/"+this.izena+".png");
			azalpena.x = 0;
			azalpena.y = 0;
			addChild(azalpena);
			azalpena.addEventListener(MouseEvent.CLICK, azalpenaKendu);
		}
		
		private function azalpenaKendu(eventObject:MouseEvent){
			removeChild(azalpena);
			
		}


	}
	
}
